# Maintainer: Antonio Rojas <arojas@archlinux.org>

pkgname=mmtf-cpp
pkgver=1.1.0
pkgrel=1
pkgdesc='The pure C++ implementation of the MMTF API, decoder and encoder'
arch=(any)
url='http://mmtf.rcsb.org/'
license=(MIT)
depends=(msgpack-cxx)
makedepends=(cmake)
source=(https://github.com/rcsb/mmtf-cpp/archive/v$pkgver/$pkgname-$pkgver.tar.gz)
sha256sums=('021173bdc1814b1d0541c4426277d39df2b629af53151999b137e015418f76c0')

build() {
  cmake -B build -S $pkgname-$pkgver \
    -DCMAKE_INSTALL_PREFIX=/usr
  cmake --build build
}

package() {
  DESTDIR="$pkgdir" cmake --install build
  install -Dm644 $pkgname-$pkgver/LICENSE -t "$pkgdir"/usr/share/licenses/$pkgname
}
